

## All Aboard! Next Stop is the Destination Excel

William E Benjamin Jr, Owl Computer Consultancy, LLC, Phoenix AZ.

### ABSTRACT

Over the last few years both Microsoft Excel file formats and the SAS® interfaces to those Excel formats have changed. SAS® has worked hard to make the interface between the two systems easier to use. Starting with “Comma Separated Variable” files and moving to PROC IMPORT and PROC EXPORT, LIBNAME processing, SQL processing, SAS® Enterprise Guide®, JMP®, and then on to the HTML and XML tagsets like MSOFFICE2K, and EXCELXP. Well, there is now a new entry into the processes available for SAS users to send data directly to Excel. This new entry into the ODS arena of data transfer to Excel is the ODS destination called EXCEL. This process is included within SAS ODS and produces native format Excel files for version 2007 of Excel and later. It was first shipped as an experimental version with the first maintenance release of SAS® 9.4. This ODS destination has many features similar to the EXCELXP tagsets.

### INTRODUCTION

Three weeks before my book [2] went to SAS Press for the final review I found out about a new experimental feature of the SAS system had been shipped with SAS 9.4 TS Level 1M1. This feature allowed the direct output of Excel files in the new \*.xlsx format. Since I was using this version of SAS, I was able to include an example in the book. Since that time the ODS EXCEL destination has been released as a supported feature of SAS available to anyone using any version of SAS that has the third maintenance release of SAS 9.4 or greater as the base product, including SAS University Edition®.

### PROBLEM

One issue that I have found over the years is that many programmers have a hard time (or take no time) searching for details of SAS coding features. Many new programmers confine themselves to cutting and pasting old code and tweaking (a technical term) for their needs today, and tomorrow, and the next day, and ... so on. Think about the programs you are writing. Do you still see code that uses “PROC IMPORT” or PROC EXPORT”? If you do, then do you also have procedures set up to manually reformat the output Excel files? The new ODS EXCEL destination may be able to make you coding easier.

### ODS DESTINATION EXCEL

All ODS commands are described in the SAS online manual [1] “SAS® 9.4 Output Delivery System: User's Guide, Fourth Edition” including the commands that relate to the EXCEL destination. I found a copy at the following SAS web site: <http://support.sas.com/documentation/cdl/en/odsug/67921/PDF/default/odsug.pdf> For this paper I have reformatted some of the information from “Chapter 6 The Dictionary of ODS Language Statements”. Specifically, the ODS EXCEL Statement section. While the final authority will always be the SAS documentation I have reformatted information on Page 292 and following to put the information into a tabular format to get as much information into the limited space available here. Because this EXCEL destination is now part of ODS, it is part of any version of SAS that has at its base the third maintenance release of SAS 9.4 or greater.

In addition, as part of ODS the EXCEL destination also works on at least some “NON-Windows” based computer systems. See the SAS documentation to see which platforms are available. One system is the z/OS operating system that is usually found on IBM mainframes. This Excel destination works when using HFS files and directories, See the manual for more information.

### OPTIONS

ODS option statements can stand independent of each other. What I mean by that is the ODS EXCEL FILE= statement and the ODS EXCEL OPTIONS statements do not need to be on the same statement. For example the following two code segments (as coded in SAS University Edition) do the same thing.

NOTE - SAS University Edition directory and file definitions vary slightly from standard SAS installation file definitions and are used throughout this paper.

```

filename aa "/folders/myfolders/Excel_File_1.xlsx";
ods excel file=aa;
ods excel options(orientation='LANDSCAPE');
ods excel options(zoom='75');
  proc print data=sashelp.shoes;
    by region;
  run;
ods excel close;
run;

/* and */

filename bb "/folders/myfolders/Excel_File_2.xlsx";
ods excel file=bb;
ods excel options(orientation='LANDSCAPE' zoom='75');
  proc print data=sashelp.shoes;
    by region;
  run;
ods excel close;
run;

```

The ODS EXCEL statement has both additional arguments and options that can be applied to build and modify the output Excel file. I have listed below additional arguments and created a table of the options. I consider most of the options to be more useful than the arguments, but acknowledge the arguments have value too. See the manual [1, page 292 et al] for more information. This paper only uses the FILE and CLOSE additional arguments. The input SAS file is the WORK.SHOES file for all examples.

- (ID= identifier)
- ANCHOR='anchor-name'
- AUTHOR='text-string'
- BOX\_SIZING=(CONTENT\_BOX | BORDER\_BOX)
- CATEGORY='text-string'
- CLOSE
- COMMENTS='text-string'
- CSSSTYLE='file-specification'<(media-type1<...media-type-10)>>
- DOM<="external-file">
- DPI='number'
- EXCLUDE exclusion(s) | ALL | NONE
- FILE='file-specification'
- GFOOTNOTE | NOGFOOTNOTE
- GTITLE | NOGTITLE
- IMAGE\_DPI='number'
- KEYWORDS='text-string'
- OPTIONS (< suboption(s)>)
- SASDATE
- SELECT selection(s) | ALL | NONE

- SHOW
- STATUS='text-string'
- STYLE= style-override(s)
- TEXT='text-string'
- TITLE='text-string'
- WORK='fileref' | 'directory-name'

The following table was adapted from “Chapter 6 The Dictionary of ODS Language Statements” [1, page 299 et al]

<b>ODS Destination EXCEL Features</b>		
Option	Default	Function
<b>ABSOLUTE_COLUMN_WIDTH</b> ='number-list'   'NONE'	NONE	Specifies the column widths. Lists widths to use for columns instead of allowing SAS to determine the column width (measured widths). The number-list is a comma separated list of numbers. You can use 'NONE' to reset to the default.
<b>ABSOLUTE_ROW_HEIGHT</b> ='number_list'	NIL	Specifies the row heights. Lists heights to use for each row instead of allowing SAS to determine the column height (measured height). You can Delimit multiple row values with commas.
<b>AUTOFILTER = 'ALL'   'NONE'   'range'</b>	NONE	Turns on filtering for specified columns in the worksheet. <ul style="list-style-type: none"> <li>• 'ALL' turns on an autofilter to all columns.</li> <li>• 'NONE' no autofiltering is applied.</li> <li>• 'Range' turns on filtering to the range of columns specified. For example, If a range of '3-5' is specified, the auto filter is applied to that range of columns.</li> </ul>
<b>BLACKANDWHITE= 'OFF'   'ON'</b> Aliases 'NO'   'YES'	OFF	Enables printing of the worksheet in black and white. <ul style="list-style-type: none"> <li>• ON prints the worksheet in black and white.</li> <li>• OFF does not print the worksheet in black and white.</li> </ul>
<b>BLANK_SHEET='string'</b>	NONE	Creates a blank worksheet with the specified name. The string is a string with a length greater than zero. This name is used in combination with a worksheet counter to create a unique name. Worksheet names can be up to 28 characters long.
<b>CENTER_HORIZONTAL= 'OFF'   'ON'</b> Aliases 'NO'   'YES'	OFF	Centers the worksheet horizontally when printing. <ul style="list-style-type: none"> <li>• ON centers the worksheet horizontally when printing.</li> <li>• OFF does not center the worksheet horizontally when printing.</li> </ul>
<b>CENTER_VERTICAL= 'OFF'   'ON'</b> Aliases 'NO'   'YES'	OFF	Specifies if the worksheet is to be centered vertically when printing. <ul style="list-style-type: none"> <li>• ON centers the worksheet vertically when printing.</li> <li>• OFF does not center the worksheet vertically when printing.</li> </ul>
<b>COLUMN_REPEAT="number "   "number-range"   "HEADER"</b>	nil (no repeating)	This controls how column headings are repeated across pages. <ul style="list-style-type: none"> <li>• HEADER repeat any of the columns containing headers.</li> </ul>

		<ul style="list-style-type: none"> <li>Number specifies that the header of the column specified is repeated on each page.</li> <li>Number-range specifies that the headers of the columns within the range specified are repeated on each page.</li> </ul>
<b>CONTENTS= 'OFF'   'ON'</b> Aliases 'NO'   'YES'	OFF	<p>Creates a worksheet that contains the table of contents.</p> <ul style="list-style-type: none"> <li>ON creates a worksheet that contains the table of contents.</li> <li>OFF creates a worksheet that does not contain a table of contents.</li> </ul>
<b>DPI='number'</b>	300 DPI	Specifies the dots per inch for print resolution. Numbers allowed are 300, 600, and 1200.
<b>DRAFTQUALITY= 'OFF'   'ON'</b> Aliases 'NO'   'YES'	OFF	<p>Specifies if draft quality should be used for printing</p> <ul style="list-style-type: none"> <li>ON specifies that draft quality should be used for printing.</li> <li>OFF specifies that draft quality should not be used for printing.</li> <li>Note Graphs will not be printed if DRAFTQUALITY =YES.</li> </ul>
<b>EMBEDDED_FOOTNOTES='OFF'   'ON'</b> Aliases 'NO'   'YES'	OFF	<p>Specifies whether footnotes should appear in the worksheet.</p> <ul style="list-style-type: none"> <li>ON embed footnotes in the worksheet.</li> <li>OFF do not embed footnotes in the worksheet.</li> </ul>
<b>EMBED_FOOTNOTES_ONCE='OFF'   'ON'</b> Aliases EMBED_FOOTNOTES_ONCE =	OFF	<p>Specifies whether embedded footnotes should appear only at the bottom of the worksheet.</p> <ul style="list-style-type: none"> <li>ON embedded footnotes appear only once at the bottom of the worksheet.</li> <li>OFF embedded footnotes appear at the bottom of the worksheet.</li> </ul>
<b>EMBEDDED_TITLES= 'OFF'   'ON'</b> Aliases 'NO'   'YES'	ON	<p>Specifies whether titles should appear in the worksheet.</p> <ul style="list-style-type: none"> <li>ON embed titles in the worksheet.</li> <li>OFF do not embed titles in the worksheet.</li> </ul>
<b>EMBED_TITLES_ONCE= 'OFF'   'ON'</b> Aliases 'NO'   'YES'	OFF	<p>Specifies whether embedded titles should appear at the top of the worksheet only once.</p> <ul style="list-style-type: none"> <li>ON embedded titles appear only once at the top of the worksheet.</li> <li>OFF titles appear as they normally appear.</li> </ul>
<b>FITTOPAGE= 'OFF'   'ON'</b> Aliases 'NO'   'YES'	OFF	<p>Specifies that the worksheet should fit on a page when printing.</p> <ul style="list-style-type: none"> <li>ON fits the worksheet on the page when printing.</li> <li>OFF does not try to fit the worksheet on the page when printing.</li> </ul>
<b>FORMULAS= 'OFF'   'ON'</b> Aliases 'NO'   'YES'	ON	<p>Specifies if data values that begin with an "=" become formulas or cell values.</p> <ul style="list-style-type: none"> <li>ON data values that begin with an "=" become formulas.</li> <li>OFF data values that begin with an "=" become cell</li> </ul>

		values.
<b>FROZEN_HEADERS= 'OFF'   'ON'   number</b> <b>Aliases 'FALSE'   'YES'</b>	OFF	Specifies that headers can scroll or not scroll with the scroll bar. <ul style="list-style-type: none"> <li>ON headers do scroll with the scroll bar.</li> <li>OFF headers do not scroll with the scroll bar.</li> <li>Number the number of the header row that does not scroll with the scroll bar.</li> </ul>
<b>FROZEN_ROWHEADERS= 'OFF'   'ON'   number</b> <b>Aliases 'NO'   'YES'</b>	OFF	Specifies if the row headers on the left scroll when the table data scrolls. <ul style="list-style-type: none"> <li>ON the header rows on the left scroll when the table data scrolls.</li> <li>OFF no headers are frozen.</li> <li>Number freeze the number of columns specified</li> </ul>
<b>GRIDLINES= 'OFF'   'ON'</b> <b>Aliases 'NO'   'YES'</b>	OFF	Specifies if grid lines are printed. <ul style="list-style-type: none"> <li>ON grid lines are printed.</li> <li>OFF grid lines are not printed.</li> </ul>
<b>HIDDEN_COLUMNS = 'number_list_range'</b>	All columns are shown.	Specifies the columns to hide. The columns identified are hidden. You can specify a number to hide a specific column, a list of numbers to hide a bunch of columns, or a range of numbers to hide consecutive columns. Note: Each value in the number_list_range is separated by commas.
<b>HIDDEN_ROWS = 'number_list_range'</b>	All rows are shown.	Specifies the rows to hide. You can specify a list of rows to hide or a range of rows to hide.
<b>INDEX= 'OFF'   'ON'</b> <b>Aliases 'NO'   'YES'</b>	OFF	Creates a worksheet that contains an index of all worksheets. <ul style="list-style-type: none"> <li>ON creates a worksheet that contains an index of all worksheets.</li> <li>OFF does not create a worksheet that contains an index of all worksheets.</li> </ul>
<b>MSG_LEVEL= 'string'</b>	NO	Suppresses messages from the Excel.
<b>ORIENTATION= 'PORTRAIT'   'LANDSCAPE'</b>	PORTRAIT	Orients the printed page as either portrait or landscape. <ul style="list-style-type: none"> <li>PORTRAIT prints a portrait-oriented page.</li> <li>LANDSCAPE prints a landscape-oriented page.</li> </ul>
<b>PAGE_ORDER_ACROSS= 'OFF'   'ON'</b>	OFF	Specifies that the information across the page is printed first followed by the information that is down the page. <ul style="list-style-type: none"> <li>ON print all of the information across the page first, followed by the information down the page.</li> <li>OFF print all of the information down the page first, followed by the information across the page.</li> </ul>
<b>PAGES_FITHEIGHT= 'number'</b>		Specifies the number of pages down to fit the worksheet when printing.
<b>PAGES_FITWIDTH= 'number'</b>		Specifies the number of pages to fit the worksheet across when printing.
<b>PRINT_AREA= 'item'</b>	NONE	Describes the printed area in terms of the column and row to start and end with. You can use column and row numbers and letters. For example,  print_area='a,2,g,20' indicates top left corner and bottom right corner. Separate each PRINT_AREA item with a comma.

<b>PRINT_FOOTER='text-string'</b>		Specifies the text that is placed in the footer when printing. If a footnote is specified, that footnote is used. Otherwise, this text is placed in the footer.
<b>PRINT_FOOTER_MARGIN='number'</b>	0.5 inches	Specifies the footer margin that is set in the page setup window when printing. This margin is measured in inches.
<b>PRINT_HEADER='text-string'</b>		Specifies the text that is placed in the header when printing. If no title is specified, this text is used by Excel on the printed page. If a title has been specified with the TITLE statement, that title is used.
<b>PRINT_HEADER_MARGIN='number'</b>	0.5 inches	Specifies the header margin that is set in the page setup dialog window when printing. This margin is measured in inches.
<b>ROWBREAKS_COUNT='number'</b>		Specifies that for every number data rows, insert a print page for printing.
<b>ROWBREAKS_INTERVAL= 'OUTPUT'   'PROC'   'NONE'</b>	NONE	Controls the placement of page breaks. This option places a page break after each output object or after each procedure. <ul style="list-style-type: none"> <li>• OUTPUT insert a page break between output objects.</li> <li>• PROC inserts a page break between each procedure's output.</li> <li>• NONE does not insert custom page breaks.</li> </ul>
<b>ROWCOLHEADINGS= 'OFF'   'ON'</b> Aliases 'NO'   'YES'	OFF	Specifies if row and column headings should be printed. <ul style="list-style-type: none"> <li>• ON prints row and column headings.</li> <li>• OFF does not print row and column headings.</li> </ul>
<b>ROW_HEIGHTS ='number_list'</b>		Specifies the height of the row. The measurement is in points and is the positional array of row height values. This value overrides the default height. The parameters of this option are positional, but not all values must be specified. A value of 0 means that the height should be taken from the style. The first value is the height for table header rows. The next is the height for the table body rows. The next value is the row height for BY lines. The fourth is for titles, the fifth is for footers, the sixth is the page break height, and the last value is the height for paragraph skip. By default, the measurement is taken from the font size. The table row height is defined by the font size in the header style.  Delimit multiple row values with commas.
<b>ROW_REPEAT='NONE'   'HEADER'   'number'   number-range'</b>	NONE	Controls how row headings are repeated across pages. For example, row_repeat="header" repeats all the row headers, row_repeat="1" repeats only one row, and row_repeat="1-3" repeats rows 1, 2, and 3. <ul style="list-style-type: none"> <li>• NONE specifies that no rows are repeated on each page.</li> <li>• HEADER specifies that all row headings are repeated on each page.</li> <li>• Number specifies that the header of the row specified is repeated on each page. Example The following example specifies that the header for row 1 is repeated across the printed page. ods excel options(row_repeat='1');</li> <li>• Number-range specifies that the headers of the rows within the range specified are repeated on each page. Example The following example specifies that the</li> </ul>

		headers for rows 1, 2, and 3 are repeated across the printed page. ods excel options(row_repeat='1-3');
<b>SCALE='number'</b>	100	Specifies the scale level for printing.
<b>SHEET_INTERVAL= 'BYGROUP'   'PAGE'   'PROC'   'NONE'   'TABLE'</b> Alias BYGROUPS	TABLE	Specifies the criteria for when a new worksheet is created. <ul style="list-style-type: none"> <li>• BYGROUP creates a new worksheet after each BYGROUP.</li> <li>• NONE creates one worksheet with all of the data.</li> <li>• PAGE creates a worksheet for each page of procedure output.</li> <li>• PROC creates a worksheet of all of the procedure output.</li> <li>• TABLE creates a worksheet for each table.</li> </ul>
<b>SHEET_LABEL='text-string'   'NONE'</b>	NONE	Used as the first part of the name in the worksheet label instead of the predefined string. This option is used in combination with the various worksheet naming options like SHEET_INTERVAL.  Note: The sheet label prepends to the sheet name. <ul style="list-style-type: none"> <li>• NONE creates worksheets named by default.</li> <li>• text-string names the first part of the label of a worksheet with the specified string.</li> </ul>
<b>SHEET_NAME='text-string'</b>		Specifies the name for the next worksheet. This name is used along with the worksheet counter to create a unique name. Range Worksheet names can be up to 28 characters long.
<b>START_AT='string'</b>	1,1	Specifies a starting cell for the report. The default is to start at column 1 and row 1. This option cannot be changed in the middle of a sheet.
<b>SUPPRESS_BYLINES= 'OFF'   'ON'</b> Aliases 'NO'   'YES'	OFF	Specifies whether to suppress BY lines in the worksheet. <ul style="list-style-type: none"> <li>• ON suppress BY lines in the worksheet.</li> <li>• OFF BY lines appear in the worksheet.</li> </ul>
<b>TAB_COLOR='string'</b>		Specifies the color for the next worksheet.  Examples: <ul style="list-style-type: none"> <li>• ods excel options (tab_color='red');</li> <li>• ods excel options (tab_color='#ff0000');</li> <li>• ods excel options (tab_color='rgba(0,100%,0,0.5)');</li> </ul>
<b>TITLE_FOOTNOTE_NOBREAK='number'</b> Aliases 'NO'   'YES'	NO	Specifies that titles and footnotes do not wrap across lines. <ul style="list-style-type: none"> <li>• ON Titles and footnotes wrap across lines.</li> <li>• OFF Titles and footnotes do not wrap across lines.</li> </ul>
<b>TITLE_FOOTNOTE_WIDTH='number'</b>	0 (Zero)	Specifies the number of columns that titles and footnotes should span. <ul style="list-style-type: none"> <li>• If zero, titles and footnotes are merged across the number of columns currently in use.</li> </ul>
<b>ZOOM='number'</b>	100	Indicates the initial zoom level on the worksheet.

Sample usage of the options follows, this can also be found in the manual [1, page 299 et. al.]

1. Example: ods excel options(absolute\_column\_width='16');
2. Example: ods excel options(absolute\_row\_height='20');

3. Example: ods excel options(autofilter='all');
4. Example: ods excel options(blackandwhite='on');
5. Example: ods excel options(blank\_sheet='SAS Sheet 1');
6. Example: ods excel options(center\_horizontal='yes');
7. Example: ods excel options(center\_vertical='yes');
8. Example: ods excel options(column\_repeat='1');
9. Example: ods excel options(contents='yes');
10. Example: ods excel options(dpi='600');
11. Example: ods excel options(draftquality='on');
12. Example: ods excel options(embedded\_footnotes='yes');
13. Example: ods excel options(embed\_footnotes\_once='yes');
14. Example: ods excel options(embedded\_titles='on');
15. Example: ods excel options(embed\_titles\_once='on');
16. Example: ods excel options(fittopage='on');
17. Example: ods excel options(formulas='off');
18. Example: ods excel options(frozen\_headers='on');
19. Example: ods excel options(frozen\_rowheaders='yes');
20. Example: ods excel options(gridlines='on');
21. Example: ods excel options(hidden\_columns='1, 2, 5, 6, 8-10');
22. Example: ods excel options(hidden\_rows='1,2,5,6,8-10');
23. Example: ods excel options(index='on');
24. Example: ods excel options(msg\_level='NO');
25. Example: ods excel options(orientation='landscape');
26. Example: ods excel options(page\_order\_across='on');
27. Example: ods excel options(pages\_fitheight='3');
28. Example: ods excel options(pages\_fitwidth='3');
29. Example: ods excel options(print\_area='a,2,g,20');
30. Example: ods excel options(print\_footer="Draft Copy for Review text");
31. Example: ods excel options(print\_footer\_margin="2");
32. Example: ods excel options(print\_header='My custom header');
33. Example: ods excel options(print\_header\_margin="1");
34. Example: ods excel options(rowbreaks\_count="20");
35. Example: ods excel options(rowbreaks\_interval='proc');
36. Example: ods excel options(rowcolheadings='on');
37. Example: ods excel options(row\_heights='20,50,100,20,50,100,50');
38. Example: ods excel options(row\_repeat='HEADER');
39. Example: ods excel options(scale='10');
40. Example: ods excel options(sheet\_interval='proc');
41. Example: ods excel options(sheet\_label='country');
42. Example: ods excel options(sheet\_name='PROC REPORT Stats');
43. Example: ods excel options(start\_at='2,2');
44. Example: ods excel options(suppress\_bylines='on');
45. Example: ods excel options(tab\_color='red');
46. Example: ods excel options(title\_footnote\_nobreak='yes' embedded\_titles='yes');

47. Example: ods excel options(title\_footnote\_width='10');

48. Example: ods excel options(zoom='75');

## CODE AND RESULTS

I cannot give examples for all of these options but I will present several options and the output EXCEL workbook results.

### EXAMPLE 1:

The following code should provide an interesting output EXCEL file. I am changing the cell sizes and setting up a landscape printout. The by statement in the PROC PRINT step caused the paging of the outputs by region to be in different worksheets. The output shown is for the by group of "Asia" but all of the by groups are in the workbook:

```
filename aa "/folders/myfolders/Excel_File_1.xlsx";
ods excel file=aa;
ods excel options (ABSOLUTE_COLUMN_WIDTH='5,10,15,20,25,30,35'
                  ABSOLUTE_ROW_HEIGHT='30,10'
                  orientation='LANDSCAPE' );
proc print data=sashelp.shoes;
  by region;
run;
ods excel close;
run;
```

Note that each column is bigger than the one to the left, and the cell heights alternate.

Obs	Product	Subsidiary	Stores	Sales	Inventory
57	Boot	Bangkok	1	\$1,996	\$9,576
58	Men's Dress	Bangkok	1	\$3,033	\$20,831
59	Sandal	Bangkok	1	\$3,230	\$15,087
60	Slipper	Bangkok	1	\$3,019	\$16,075
61	Women's Ca	Bangkok	1	\$5,389	\$16,251
62	Boot	Seoul	17	\$60,712	\$160,589
63	Men's Casue	Seoul	1	\$11,754	\$2,176
64	Men's Dress	Seoul	7	\$116,333	\$251,803
65	Sandal	Seoul	3	\$4,978	\$21,483
66	Slipper	Seoul	21	\$149,013	\$469,007
67	Sport Shoe	Seoul	1	\$937	\$455
68	Women's Ca	Seoul	2	\$20,448	\$36,576
69	Women's Dri	Seoul	7	\$78,234	\$140,628
70	Sport Shoe	Tokyo	1	\$1,155	\$15,602

Figure 1 – Output using ABSOLUTE\_COLUMN\_WIDTH and ABSOLUTE\_ROW\_HEIGHT options.

And the output orientation is landscape not the default of Portrait.




Figure 2 – Excel PAGE SETUP page showing LANDSCAPE setting.

### EXAMPLE 2:

In this example I am applying an “Auto Filter” to three columns, hiding two columns, turning the tabs “RED”, and applying a printout mode of landscape to the output EXCEL workbook. This still has individual sheets for each “BY” group with the “Asia” group shown.

```
filename bb "/folders/myfolders/Excel_File_2.xlsx";
ods excel file=bb;
ods excel options(autofilter='1-3'
                 hidden_columns='5,6'
                 tab_color='red'
                 orientation='LANDSCAPE' );
proc print data=sashelp.shoes;
  by region;
run;
ods excel close;
run;
```

Notice here that columns “E” and “F” are hidden and columns “A”, “B”, and “C” are filtered. When you open the workbook the tabs may be shown as shaded with a color gradient. When you select a tab the others change to a solid color.




Figure 3 – Excel output sheet shown with hidden columns and auto-filter on some columns.

**EXAMPLE 3:**

My third example adds a little flare to the output by adding titles and footnotes to the Worksheet pages by adding titles and footnotes that are common on every page.

```
filename bb "/folders/myfolders/Excel_File_3.xlsx";
Title "Listing of the Shoes Dataset By Region";
Footnote "Data Shown by Region";
ods excel file=bb;
ods excel options(EMBEDDED_TITLES='ON'
                  EMBED_FOOTNOTES_ONCE='ON'
                  FROZEN_HEADERS='ON'
                  scale='110');
proc print data=sashelp.shoes;
  by region;
run;
ods excel close;
run;
```

This output workbook also shows the scale percentage as being 110% and to highlight a point I scrolled the EXCEL worksheet down to verify that the header rows were frozen. Notice the EXCEL row counter skips from 5 to 15. This is the same data as shown in examples 1 and 2.

Obs	Product	Subsidiary	Stores	Sales	Inventory	Returns
66	Slipper	Seoul	21	\$149,013	\$469,007	\$2,941
67	Sport Shoe	Seoul	1	\$937	\$455	\$10
68	Women's Casual	Seoul	2	\$20,448	\$36,576	\$790
69	Women's Dress	Seoul	7	\$78,234	\$140,628	\$1,891
70	Sport Shoe	Tokyo	1	\$1,155	\$15,602	\$22

Figure 4 – Excel one of several output sheets showing formatted output.

#### EXAMPLE 4:

In this example the worksheet names are changed and the output is shifted to start in row “3” and column “C”.

```
filename bb "/folders/myfolders/Excel_File_4.xlsx";
Title "Listing of the Shoes Dataset By Region";
Footnote "Data Shown by Region";
ods excel file=bb;
ods excel options(EMBEDDED_TITLES='ON'
                  EMBED_FOOTNOTES_ONCE='ON'
                  FROZEN_HEADERS='ON'
                  SUPPRESS_BYLINES='ON'
                  SHEET_LABEL='Shoes'
                  Start_at='3,3');
proc print data=sashelp.shoes;
  by region;
run;
ods excel close;
run;
```

Also notice that the BY line header “REGION=Asia” is missing on this page (all pages really), and the headers are still frozen.

Obs	Product	Subsidiary	Stores	Sales	Inventory	Returns
63	Men's Casual	Seoul	1	\$11,754	\$2,176	\$833
64	Men's Dress	Seoul	7	\$116,333	\$251,803	\$2,443
65	Sandal	Seoul	3	\$4,978	\$21,483	\$105
66	Slipper	Seoul	21	\$149,013	\$469,007	\$2,941
67	Sport Shoe	Seoul	1	\$937	\$455	\$10
68	Women's Casual	Seoul	2	\$20,448	\$36,576	\$790
69	Women's Dress	Seoul	7	\$78,234	\$140,628	\$1,891
70	Sport Shoe	Tokyo	1	\$1,155	\$15,602	\$22

Figure 5 – Excel one of several output sheets showing formatted output offset from row 1 column 1.

## CONCLUSION

I hope you have figured out by now that this new ODS destination called “EXCEL” is a powerful addition to the SAS system. The most important part of the tool is that it is part of the Output Delivery System (ODS) of Base SAS and creates EXCEL output files in the native EXCEL format for \*.xlsx files. Additionally you can now write the native format EXCEL workbooks on non-windows computer systems.

## REFERENCES

[1] SAS Institute Inc. 2015. SAS® 9.4 Output Delivery System: User's Guide, Fourth Edition. Cary, NC: SAS Institute Inc.

[2] Benjamin, William E., Jr. 2015. *Exchanging Data Between SAS® and Microsoft Excel: Tips and Techniques to Transfer and Manage Data More Efficiently*, Cary, NC: SAS Institute Inc.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: William E Benjamin Jr  
 Enterprise: Owl Computer Consultancy, LLC  
 Address: P.O.Box 42434  
 City, State ZIP: Phoenix AZ, 85080  
 Work Phone: 623-337-0269  
 E-mail: William@owlcomputerconsultancy.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.